

# PROC REPORT: AN INTRODUCTION TO THE BATCH LANGUAGE

Ray Pass  
ASG, Inc.

In 1990, the SAS Institute included the experimental release of the REPORT procedure with version 6.06 of the SAS System. This was followed in 1991 with various alpha/beta releases of REPORT, and a 1992 production release in version 6.07. PROC REPORT is a powerful replacement (addition) for PROC PRINT, and it has the full support of SAS Institute, which was lacking with the ill-fated PROC QPRINT. The QPRINT procedure could also have inherited PROC PRINT's role as the main report writing tool of the SAS\* System, but major development efforts were thrown instead behind the REPORT procedure, with its raft of new features. The result is PROC REPORT, a powerful new report writer which is certainly destined to become a SAS System productivity workhorse.

The main thrust behind PROC REPORT has been its interactive WYSIWYG ability, in which reports are designed online via screen-painting manipulation of actual output. When the desired result is achieved, the design and structure of the report can be saved as a catalog entry and called up repeatedly for additional runs with new sets of data. The design can then be modified at any time, and can either be replaced or saved as an additional entry. This is all done in a Display Manager-like environment, and will certainly attract users comfortable with working interactively. Although the batch language of PROC REPORT is also available with version 6.07 or later (not so with earlier experimental releases, or at least not easily available), it has not received the acclaim or publicity that is afforded the interactive product. In actual fact however, the batch language is every bit as powerful and feature-laden as its online cousin, and is an almost mirror-image.

The purpose of this paper is to introduce the batch version of PROC REPORT through a progressive series of annotated examples. It is not meant to be a full introductory text, and is therefore not laid out in a structured manual format with complete lookup and reference functionality. The full documentation for batch PROC REPORT can be found in SAS Technical Report P-258, Using the REPORT Procedure in a Nonwindowing Environment. The procedure is still being developed and modified however, and there are features which are not included in the manual, but which are included in the 6.07 and 6.08 enhancement documents (Tech Reports P-222 and P-242 respectively.) Future documentation releases will undoubtedly be more complete, but the presently

available manual is a fully adequate beginning.

The data set we will use throughout this paper is a fictitious academic survey wherein 100 college students were questioned at random about their spending habits during a typical year at school. **LISTING-1** presents a description of data set REPORT.ACAD. This work was done in an MVS environment, but the REPORT syntax is identical on all platforms on which it is presently implemented. Note that in version 6 of the SAS System, external data sets may be called directly in an INFILE statement. Two FORMATS are being stored permanently for further use throughout. The initial PROC REPORT call as displayed in the last line of code in **LISTING-1** produces the vanilla PROC PRINT-like display found in **OUTPUT-1** (the entire 100 observation data set is not shown in any of the non-grouped outputs).

```
DATA REPORT.ACAD;
  INFILE 'PLIB.IC.REPORT(REPDATA)';

  INPUT  @01 FNAME      $CHAR10.
         @12 LINIT     $CHAR1.
         @15 SEX       $CHAR1.
         @18 ACADYEAR  $CHAR1.
         @21 GPA       3.1
         @27 AGE       2.
         @31 BOOKS    4.
         @40 FOOD     4.
         @49 ENT      4. ;

*-----;
PROC FORMAT LIBRARY=LIBRARY;
  VALUE $SEXFMT 'M' = 'MALE'
           'F' = 'FEMALE';
  VALUE $YEARFMT '1' = 'FRESHMAN'
              '2' = 'SOPHOMORE'
              '3' = 'JUNIOR'
              '4' = 'SENIOR';
*-----;
PROC REPORT DATA=REPORT.ACAD;
```

## LISTING-1

As **OUTPUT-1** shows, PROC REPORT in its simplest format is a PROC PRINT look-alike. **LISTING-2** produces the report found in **OUTPUT-2**, and introduces the basic features of REPORT syntax. The first items to note are the HEADLINE and HEADSKIP options on the PROC REPORT statement. These are two of the many formatting and structural options available on this statement, and produce the underline and blank line under the column headings. The COLUMNS statement declares the variables to be included in the report structure and the order in which they are to be processed from left to right. There are options available on the COLUMNS

statement, one of which is described below. Note that there are two variables, NAME and TOTAL, which are not found in the input data set, but which are rather created for use in the report via the COMPUTE statements described below. These variables are transient and exist only for the duration of the current REPORT run; they are not added to the dataset being REPORT'ed. The series of DEFINE statements is used to assign formatting information to the variables. The first two, for NAME and LINIT, instruct REPORT not to print these two variables, via a NOPRINT option. They must be included in the COLUMNS statement because the NAME variable is computed from them. They are therefore part of the data set being REPORT'ed, but are not included in the output. Order is important here as NAME must be preceded by the variables it is computed from.

```

PROC REPORT DATA=REPORT.ACAD HEADLINE
HEADSKIP;

COLUMNS FNAME LINIT NAME SEX ACADYEAR
GPA AGE BOOKS FOOD ENT TOTAL;

DEFINE FNAME / DISPLAY NOPRINT;
DEFINE LINIT / DISPLAY NOPRINT;
DEFINE SEX / DISPLAY FORMAT=$SEXFMT.
WIDTH=6;
DEFINE ACADYEAR / DISPLAY FORMAT=$YEAREMT.
WIDTH=9 'ACADEMIC/YEAR';
DEFINE GPA / ANALYSIS FORMAT=3.1
WIDTH=3;
DEFINE AGE / ANALYSIS FORMAT=3.
WIDTH=3;
DEFINE BOOKS / ANALYSIS FORMAT=DOLLAR6.
WIDTH=7 '$-BOOKS';
DEFINE FOOD / ANALYSIS FORMAT=DOLLAR6.
WIDTH=6 '$-FOOD';
DEFINE ENT / ANALYSIS FORMAT=DOLLAR6.
WIDTH=11 '$-ENTERTAIN';
DEFINE NAME / COMPUTED WIDTH=10;
DEFINE TOTAL / COMPUTED FORMAT=DOLLAR6.
WIDTH=7 '$-TOTAL';

COMPUTE NAME /CHAR LENGTH=10;
NAME=TRIM(FNAME) || ' ' || LINIT || '.';
ENDCOMP;

COMPUTE TOTAL;
TOTAL = BOOKS.SUM + FOOD.SUM + ENT.SUM;
ENDCOMP;

```

**LISTING-2**

PROC REPORT supports six different types of output variables with different uses: DISPLAY, ANALYSIS, ORDER, GROUP, ACROSS and COMPUTED. In this example, the DEFINE statements declare NAME, LINIT, SEX and ACADYEAR as DISPLAY variables, GPA, AGE, BOOKS, FOOD and ENT as ANALYSIS variables, and NAME and TOTAL as COMPUTED variables. Character variables are DISPLAY and numeric variables

are ANALYSIS by default, but it is good practice to include these designations anyway; it makes for ease of reading and modification of the code. Other variable types are described below. DISPLAY and ANALYSIS variables are listed in the output, one printed value per observation. The FORMAT and WIDTH options declare output value formats and column widths, and the quoted strings create the column headings. Note that ACADYEAR has a split heading, and that the WIDTHs must be adequate to contain the headings ( e.g. \$-ENTERTAIN needs a WIDTH=11).

The variables NAME and TOTAL were included in the COLUMNS statement, and declared in the DEFINE statements as COMPUTED variables. These variables are created in 'code segments' which each begin with a COMPUTE statement and end with an ENDCOMP statement. The code segments contain DATA step statements which create the COMPUTED variables. Options on the COMPUTE statement which create NAME declare it to be a CHARACTER variable with a LENGTH of 10. The compound name syntax for the variables in the TOTAL compute statement determine which statistic is used (in this case, SUM.)

**OUTPUT-2** displays a useful, formatted, one line per observation report as created with the code in **LISTING-2**. Much can be done however in terms of ordering the data for presentation. **LISTING-3** contains the code which produces the categorized and sorted data displayed in **OUTPUT-3**. Here the data is ordered by ACADYEAR and SEX, and presented with only one printing each of the unique values of ACADYEAR and SEX, when they change.

The first change in the code from the previous listing to note is that the order of variables in the COLUMNS statement has been changed. Variables used to order the data must be processed before (to the left of) any DISPLAY variable. This makes intuitive sense as you look at the output, and is backed up by rigid syntactical placement rules. The DEFINE statements declare ACADYEAR and SEX to be ORDER variables. This is what actually creates the sorted display. Note that although the sequenced order of variables is critical in the COLUMNS statement, it is of no concern in the series of DEFINE statements. The DEFINE statement for ACADYEAR also includes the ORDER= option, which in this case instructs REPORT to order the observations sorted by INTERNAL values (1, 2, 3, 4). This eliminates the need for a prior sorting of the data set by a PROC SORT. Other possible presentation orders

(FORMATTED, DATA, FREQUENCY) are available as they are in the MEANS, FREQ, and TABULATE procedures. This feature of REPORT is undocumented in the first edition of the REPORT manual, as are others, but is included in the later version documentation.

```

PROC REPORT DATA=REPORT.ACAD HEADLINE
HEADSKIP;

COLUMNS ACADYEAR SEX FNAME LINIT NAME
GPA AGE BOOKS FOOD ENT TOTAL;

DEFINE FNAME / DISPLAY NOPRINT;
DEFINE LINIT / DISPLAY NOPRINT;
DEFINE SEX / ORDER FORMAT=$SEXFMT.
WIDTH=6;
DEFINE ACADYEAR / ORDER FORMAT=$YEARFMT.
WIDTH=9 'ACADEMIC/YEAR'
ORDER=INTERNAL;
DEFINE GPA / ANALYSIS FORMAT=3.1
WIDTH=3;
DEFINE AGE / ANALYSIS FORMAT=3.
WIDTH=3;
DEFINE BOOKS / ANALYSIS FORMAT=DOLLAR6.
WIDTH=7 '$-BOOKS';
DEFINE FOOD / ANALYSIS FORMAT=DOLLAR6.
WIDTH=6 '$-FOOD';
DEFINE ENT / ANALYSIS FORMAT=DOLLAR6.
WIDTH=11 '$-ENTERTAIN';
DEFINE NAME / COMPUTED WIDTH=10;
DEFINE TOTAL / COMPUTED FORMAT=DOLLAR6.
WIDTH=7 '$-TOTAL';

COMPUTE NAME / CHAR LENGTH=10;
NAME=TRIM(FNAME) || ' ' || LINIT || '.';
ENDCOMP;

COMPUTE TOTAL;
TOTAL = BOOKS.SUM + FOOD.SUM + ENT.SUM;
ENDCOMP;

```

**LISTING-3**

The next major feature of PROC REPORT discussed herein is the ability to collapse observations sharing the same value of any variable, or set of variables, and to display summary statistics for each subset of observations. In LISTING-4, ACADYEAR and SEX are changed to GROUP type variables, with the results displayed in OUTPUT-4.

FNAME, LINIT, and NAME are character DISPLAY variables, and cannot be meaningfully collapsed (they can be changed to ORDER or GROUP variables, but cannot remain in the design as DISPLAY variables with GROUP variables to their left). They are therefore removed from the COLUMNS and DEFINE statements, and thus from the REPORT design. The remaining variables, including the COMPUTED variable TOTAL, are summed over each ACADYEAR-SEX combination. The resulting sum is displayed in each cell of the report. Although this statistic

can be meaningful for the money variables (BOOKS, FOOD, ENT, TOTAL), a mean value would present more useful information for each subgroup. The SUM statistic is however meaningless for GPA and AGE. In addition, some formatting changes are desired. LISTING-5 contains many changes and additions which create the much enhanced display found in OUTPUT-5.

```

PROC REPORT DATA=REPORT.ACAD HEADLINE
HEADSKIP;

COLUMNS ACADYEAR SEX GPA AGE BOOKS FOOD
ENT TOTAL;

DEFINE SEX /GROUP FORMAT=$SEXFMT.
WIDTH=6;
DEFINE ACADYEAR /GROUP FORMAT=$YEARFMT.
WIDTH=9 'ACADEMIC/YEAR'
ORDER=INTERNAL;
DEFINE GPA /ANALYSIS FORMAT=3.1
WIDTH=3;
DEFINE AGE /ANALYSIS FORMAT=3.
WIDTH=3;
DEFINE BOOKS /ANALYSIS FORMAT=DOLLAR6.
WIDTH=7 '$-BOOKS';
DEFINE FOOD /ANALYSIS FORMAT=DOLLAR6.
WIDTH=6 '$-FOOD';
DEFINE ENT /ANALYSIS FORMAT=DOLLAR6.
WIDTH=11 '$-ENTERTAIN';
DEFINE TOTAL /COMPUTED FORMAT=DOLLAR6.
WIDTH=7 '$-TOTAL';

COMPUTE TOTAL;
TOTAL=BOOKS.SUM + FOOD.SUM + ENT.SUM;
ENDCOMP;

```

**LISTING-4**

The DEFINE statements for the ANALYSIS variables and the COMPUTED variable TOTAL are changed in various ways. REPORT is now instructed to display each of the ANALYSIS variables as a MEAN instead of the default SUM. In addition, the column headings for GPA and AGE are changed, and the WIDTHs are lengthened to accommodate the new headings. The COMPUTED variable TOTAL is now calculated from the new MEAN statistics of each of its components. This is done by using the compound form of each variable, e.g. BOOKS.MEAN. A column spanning heading is created in the COLUMNS statement for the money variables by including a quoted string in parentheses preceding the columns it is to span. In this heading, the initial and final characters can be extended to the total width of the set of columns it spans.

Another enhancement to the new display is the addition of summary break lines to the report. The BREAK command inserts a break line AFTER each new value of ACADYEAR. Each new line has a summary statistic (SUMMARIZE) with an overline (OL), an underline (UL) and a blank line (SKIP) before the next

printed line. The RBREAK command instructs REPORT to print a break line AFTER the entire report, with double over- and underlining (DOL, DUL) and overall summary statistics (SUMMARIZE). Each of the break commands also uses the SUPPRESS option to prevent printing break information (values, under- and overlining) under the GROUP variable columns (ACADYEAR, SEX). The last enhancements to the code in LISTING-5 is the addition of TITLE statements. These statements are standard SAS syntax, and are not saved with the REPORT definition if it is stored as a catalog entry.

```

PROC REPORT DATA=REPORT.ACAD HEADLINE
HEADSKIP;

COLUMNS ACADYEAR SEX GPA AGE
(' -ANNUAL MEAN AMOUNT SPENT ON-'
BOOKS FOOD ENT TOTAL);

DEFINE SEX / GROUP FORMAT=$SEXFMT.
WIDTH=6;
DEFINE ACADYEAR / GROUP FORMAT=$YEARFMT.
WIDTH=9 'ACADEMIC/YEAR'
ORDER=INTERNAL;
DEFINE GPA / ANALYSIS FORMAT=3.1
WIDTH=4 'MEAN/GPA' MEAN;
DEFINE AGE / ANALYSIS FORMAT=3.
WIDTH=4 'MEAN/AGE' MEAN;
DEFINE BOOKS / ANALYSIS FORMAT=DOLLAR6.
WIDTH=7 '$-BOOKS' MEAN;
DEFINE FOOD / ANALYSIS FORMAT=DOLLAR6.
WIDTH=6 '$-FOOD' MEAN;
DEFINE ENT / ANALYSIS FORMAT=DOLLAR6.
WIDTH=11 '$-ENTERTAIN'
MEAN;
DEFINE TOTAL / COMPUTED FORMAT=DOLLAR6.
WIDTH=7 '$-TOTAL';

COMPUTE TOTAL;
TOTAL=BOOKS.MEAN + FOOD.MEAN + ENT.MEAN;
ENDCOMP;

BREAK AFTER ACADYEAR / OL UL SKIP
SUMMARIZE SUPPRESS;

RBREAK AFTER / DOL DUL SKIP
SUMMARIZE;

TITLE1 'PROC REPORT - SAMPLE REPORTS';
TITLE2 '----- ACADEMIC SURVEY -----';
TITLE3;

```

LISTING-5

OUTPUT-5 presents a useful display of grouped information. PROC REPORT also allows other types of useful formatted output, both grouped and non-grouped. LISTING-6 contains code used to display the telephone-book-column style display found in OUTPUT-6.

In LISTING-6, the REPORT statement contains three new options which instruct the procedure to display

the data in three repeating 'snaking' sets of variables (PANELS) with a pagesize (PS) of 50 lines, and 10 spaces between each panel (PSPACE). Only NAME, ACADYEAR and AGE are printed (note that FNAME and LINIT must be included in COLUMNS for processing, but are NOPRINT'ed).

```

PROC REPORT DATA=REPORT.ACAD HEADLINE
HEADSKIP PANELS=3 PS=50 PSPACE=10;

COLUMNS FNAME LINIT NAME ACADYEAR GPA;

DEFINE FNAME / DISPLAY NOPRINT;
DEFINE LINIT / DISPLAY NOPRINT;
DEFINE ACADYEAR / DISPLAY WIDTH=9
FORMAT=$YEARFMT.
'ACADEMIC/YEAR';
DEFINE GPA / ANALYSIS FORMAT=3.1
WIDTH=3;
DEFINE NAME / COMPUTED WIDTH=10;

COMPUTE NAME / CHAR LENGTH=10;
NAME=TRIM(FNAME)||' '||LINIT||'.';
ENDCOMP;

TITLE1 'PROC REPORT - SAMPLE REPORTS';
TITLE2 '----- ACADEMIC SURVEY -----';
TITLE3;

```

LISTING-6

LISTING-7 also contains code for a non-grouped data display (OUTPUT-7). In this example, the WRAP option in the REPORT statement causes the values for all variables to be printed in a line-wrap fashion before printing of the next observation is begun. The NAMED option causes each variable to be preceded by its heading (variable names would be used if no headings were present) and '='. The LS (linesize) option causes REPORT to restrict horizontal printing to 70 columns. Some of the DEFINE statements use a SPACING option to manipulate the number of blank spaces printed before each value. This was done to create the quasi-columnar alignment which makes for easier reading. Each observation is also followed by a blank line before the next one is started. This was done with the aid of some pre-processing of the dataset before the REPORT procedure was called. Dataset ACAD2 was created by SET'ing REPORT.ACAD, and then creating a new variable, N, by assigning it the value of the automatic variable \_N\_. In the REPORT code, N is DEFINE'd as an ORDER variable, and a BREAK statement is used to SKIP a blank line after each new value of N. The procedure was run against ACAD2.

```

DATA ACAD2; SET REPORT.ACAD;
  N = _N_;
*-----;
PROC REPORT DATA=ACAD2 HEADLINE HEADSKIP
  NAMED WRAP LS=70;

  COLUMNS N FNAME LINIT NAME ACADYEAR AGE
  SEX GPA BOOKS FOOD ENT TOTAL;

  DEFINE N / ORDER NOPRINT;
  DEFINE FNAME / DISPLAY NOPRINT;
  DEFINE LINIT / DISPLAY NOPRINT;
  DEFINE SEX / DISPLAY FORMAT=$SEXFMT.
  WIDTH=6;
  DEFINE ACADYEAR / DISPLAY WIDTH=9 'YEAR'
  FORMAT=$YEARFMT.;
  DEFINE GPA / ANALYSIS FORMAT=3.1
  WIDTH=3;
  DEFINE AGE / ANALYSIS FORMAT=3.
  WIDTH=3;
  DEFINE BOOKS / ANALYSIS FORMAT=DOLLAR6.
  WIDTH=6 '$-BOOKS';
  DEFINE FOOD / ANALYSIS FORMAT=DOLLAR6.
  WIDTH=6 '$-FOOD'
  SPACING=3;
  DEFINE ENT / ANALYSIS FORMAT=DOLLAR6.
  WIDTH=6 '$-ENTERTAIN'
  SPACING=3;
  DEFINE NAME / COMPUTED WIDTH=10;
  DEFINE TOTAL / COMPUTED FORMAT=DOLLAR6.
  WIDTH=7 '$-TOTAL'
  SPACING=3;

  COMPUTE NAME /CHAR LENGTH=10;
  NAME=TRIM(FNAME)||' '||LINIT||'. ';
  ENDCOMP;

  COMPUTE TOTAL;
  TOTAL=BOOKS.SUM + FOOD.SUM + ENT.SUM;
  ENDCOMP;

  BREAK AFTER N / SKIP;

TITLE1 'PROC REPORT - SAMPLE REPORTS';
TITLE2 '----- ACADEMIC SURVEY -----';
TITLE3;

```

#### LISTING-7

The final example is shown in **OUTPUT-8**, and was created with the PROC REPORT code found in **LISTING-8**. This example demonstrates the last variable type allowed in PROC REPORT. The variables included in the example are ACADYEAR, SEX and GPA, and SEX is DEFINE'd as an ACROSS variable. This causes its values to be grouped like a GROUP variable, but displayed horizontally rather than vertically. The last enhancement is in the format of the COLUMNS statement. The order of processing the variable is: ACADYEAR SEX GPA. The ',' in the statement after SEX instructs REPORT to print the summary values of the ANALYSIS variable GPA under the unique values of the ACROSS variable SEX (if no statistics were included for GPA, the frequency of observations in each cell would be printed).

This syntax allows PROC REPORT to display its data in a simple PROC TABULATE-like manner, and demonstrates much of the power of this new procedure.

```

PROC REPORT DATA=REPORT.ACAD HEADLINE
  HEADSKIP;

  COLUMNS ACADYEAR SEX,GPA;

  DEFINE SEX / ACROSS FORMAT=$SEXFMT.
  WIDTH=6;
  DEFINE ACADYEAR / GROUP FORMAT=$YEARFMT.
  WIDTH=9 'ACADEMIC/YEAR';
  DEFINE GPA / ANALYSIS FORMAT=3.1
  WIDTH=6 MEAN;

TITLE1 'PROC REPORT - SAMPLE REPORTS';
TITLE2 '----- ACADEMIC SURVEY -----';
TITLE3;

```

#### LISTING-8

There are additional features in the REPORT procedure which have not been included in this paper, such as the ability to include strings of literal text in the body of the output, and the ability to conditionally include whole segments of output based on values of dataset or COMPUTED variables. One new feature worthy of note is the FLOW option on the DEFINE statement. This option will cause a long character variable to "wrap" within its column width. Since the option will honor "split characters" embedded in the text, strings can be made to break and left justify on multiple lines. This can be used to include name and address lines, for example, in a column. Other new features are being developed and will be included in future releases of the SAS System base product, and its accompanying documentation.

The REPORT procedure is a powerful and productive new tool in the SAS user's report-writing arsenal, and should be fully examined and exploited.

\*SAS is a registered trademark of SAS Institute, Inc., Cary, NC, USA.

Ray Pass, Ph.D.  
 ASG, Inc.  
 5 Sinclair Place  
 Hartsdale, NY 10530

Voice: (203) 356-9540  
 Fax: (203) 967-8644  
 e-mail: raypass@worldnet.att.net

The SAS System

	A C A D E M I C Y E A R		GPA	AGE	BOOKS	FOOD	ENT
FNAME	T X R						
DORY	A F 3		4	20	381	945	1144
GERALD	A M 3		3.5	20	265	655	687
HARRY	A M 4		3.6	19	544	855	478
REGGIE	A F 3		3.6	20	100	841	654
VALERIE	A F 3		3.2	20	532	756	250
ANTHONY	B M 1		2	16	425	855	965
GEORGE	B M 3		2.5	21	522	752	425
TOM	B M 2		3.8	19	215	625	622
CAROL	C F 4		1.6	21	344	1200	602
EDWARD	C M 1		2.9	18	451	466	2101
KATHY	C F 4		2.5	21	544	842	874
BILL	D M 3		3.8	18	244	810	620
CARMEN	D F 4		3.9	18	522	1022	587
DONNA	D F 1		3.2	19	652	802	785
FRANK	D M 4		3.6	21	675	841	254
JOAN	D F 2		2	27	680	789	855
LEON	D M 1		2.9	18	201	755	400
ROSE	D F 2		3.6	18	205	866	100
TOM	D M 2		3.6	18	406	485	544
ALLAN	E M 3		3.5	21	345	455	852
BETTY	E F 2		3.6	20	222	755	962
DENNIS	E M 3		2.8	19	541	480	221
EVELYN	E F 2		3.2	19	265	855	452
JONAH	E M 4		3.2	17	569	785	457
LILLY	E F 1		3.6	18	489	966	954
MARIA	E F 3		3.4	20	566	485	656
SAM	E M 1		3.4	18	894	688	865

OUTPUT - 1

The SAS System

NAME	SEX	ACADEMIC YEAR	GPA	AGE	\$-BOOKS	\$-FOOD	\$-ENTERTAIN	\$-TOTAL
DORY A.	FEMALE	JUNIOR	4.0	20	\$381	\$945	\$1,144	\$2,470
GERALD A.	MALE	JUNIOR	3.5	20	\$265	\$655	\$687	\$1,607
HARRY A.	MALE	SENIOR	3.6	19	\$544	\$855	\$478	\$1,877
REGGIE A.	FEMALE	JUNIOR	3.6	20	\$100	\$841	\$654	\$1,595
VALERIE A.	FEMALE	JUNIOR	3.2	20	\$532	\$756	\$250	\$1,538
ANTHONY B.	MALE	FRESHMAN	2.0	16	\$425	\$855	\$965	\$2,245
GEORGE B.	MALE	JUNIOR	2.5	21	\$522	\$752	\$425	\$1,699
TOM B.	MALE	SOPHOMORE	3.8	19	\$215	\$625	\$622	\$1,462
CAROL C.	FEMALE	SENIOR	1.6	21	\$344	\$1,200	\$602	\$2,146
EDWARD C.	MALE	FRESHMAN	2.9	18	\$451	\$466	\$2,101	\$3,018
KATHY C.	FEMALE	SENIOR	2.5	21	\$544	\$842	\$874	\$2,260
BILL D.	MALE	JUNIOR	3.8	18	\$244	\$810	\$620	\$1,674
CARMEN D.	FEMALE	SENIOR	3.9	18	\$522	\$1,022	\$587	\$2,131
DONNA D.	FEMALE	FRESHMAN	3.2	19	\$652	\$802	\$785	\$2,239
FRANK D.	MALE	SENIOR	3.6	21	\$675	\$841	\$254	\$1,770
JOAN D.	FEMALE	SOPHOMORE	2.0	27	\$680	\$789	\$855	\$2,324
LEON D.	MALE	FRESHMAN	2.9	18	\$201	\$755	\$400	\$1,356
ROSE D.	FEMALE	SOPHOMORE	3.6	18	\$205	\$866	\$100	\$1,171
TOM D.	MALE	SOPHOMORE	3.6	18	\$406	\$485	\$544	\$1,435
ALLAN E.	MALE	JUNIOR	3.5	21	\$345	\$455	\$852	\$1,652
BETTY E.	FEMALE	SOPHOMORE	3.6	20	\$222	\$755	\$962	\$1,939
DENNIS E.	MALE	JUNIOR	2.8	19	\$541	\$480	\$221	\$1,242
EVELYN E.	FEMALE	SOPHOMORE	3.2	19	\$265	\$855	\$452	\$1,572
JONAH E.	MALE	SENIOR	3.2	17	\$569	\$785	\$457	\$1,811
LILLY E.	FEMALE	FRESHMAN	3.6	18	\$489	\$966	\$954	\$2,409
MARIA E.	FEMALE	JUNIOR	3.4	20	\$566	\$485	\$656	\$1,707
SAM E.	MALE	FRESHMAN	3.4	18	\$894	\$688	\$865	\$2,447

OUTPUT - 2

The SAS System

ACADEMIC YEAR	SEX	NAME	GPA	AGE	\$-BOOKS	\$-FOOD	\$-ENTERTAIN	\$-TOTAL
FRESHMAN	FEMALE	DONNA D.	3.2	19	\$652	\$802	\$785	\$2,239
		LILLY E.	3.6	18	\$489	\$966	\$954	\$2,409
		JANE K.	2.9	18	\$402	\$1,300	\$541	\$2,243
		JACKIE M.	2.1	16	\$402	\$1,200	\$454	\$2,056
		JOAN O.	2.8	18	\$306	\$522	\$777	\$1,605
		DEBBIE R.	3.8	18	\$432	\$230	\$556	\$1,218
		SALLY R.	3.8	18	\$524	\$600	\$400	\$1,524
		ERICA S.	4.0	18	\$560	\$455	\$201	\$1,216
		ARLENE T.	3.5	18	\$512	\$466	\$412	\$1,390
		ANTHONY B.	2.0	16	\$425	\$855	\$965	\$2,245
	MALE	EDWARD C.	2.9	18	\$451	\$466	\$2,101	\$3,018
		LEON D.	2.9	18	\$201	\$755	\$400	\$1,356
		SAM E.	3.4	18	\$894	\$688	\$865	\$2,447
		JOHN F.	3.8	20	\$512	\$741	\$905	\$2,158
		JOEL H.	2.5	24	\$524	\$987	\$950	\$2,461
		MIKE J.	3.2	19	\$254	\$522	\$656	\$1,432
		FRANK L.	3.5	19	\$478	\$946	\$125	\$1,549
		GEORGE L.	2.8	18	\$490	\$954	\$795	\$2,239
		JIM R.	3.7	38	\$452	\$866	\$555	\$1,873
		ROBERT S.	3.8	17	\$500	\$547	\$220	\$1,267
SOPHOMORE	FEMALE	MANNY V.	3.4	34	\$325	\$862	\$544	\$1,731
		JOAN D.	2.0	27	\$680	\$789	\$855	\$2,324
		ROSE D.	3.6	18	\$205	\$866	\$100	\$1,171
		BETTY E.	3.6	20	\$222	\$755	\$962	\$1,939
		EVELYN E.	3.2	19	\$265	\$855	\$452	\$1,572
		CHERYL F.	3.8	19	\$245	\$577	\$622	\$1,444
		CINDY F.	3.7	19	\$318	\$622	\$455	\$1,395
		LAURIE F.	3.2	19	\$345	\$844	\$650	\$1,839
		CONNIE K.	2.9	19	\$156	\$544	\$999	\$1,699
		MARIE K.	3.2	19	\$523	\$684	\$787	\$1,994
	MALE	ELLEN T.	3.0	19	\$241	\$622	\$956	\$1,819
		MILLY T.	4.0	28	\$325	\$876	\$557	\$1,758
		ROBIN Y.	3.5	20	\$345	\$788	\$454	\$1,587
		DINA Z.	3.5	18	\$255	\$655	\$445	\$1,355
		TOM B.	3.8	19	\$215	\$625	\$622	\$1,462
		TOM D.	3.6	18	\$406	\$485	\$544	\$1,435
		WILLIAM E.	3.7	19	\$298	\$854	\$654	\$1,806
		OSCAR F.	1.5	19	\$429	\$568	\$965	\$1,962
		WAYNE G.	3.3	19	\$201	\$548	\$425	\$1,174
		ED H.	4.0	19	\$320	\$950	\$562	\$1,832
JUNIOR	FEMALE	WILLIAM I.	3.6	20	\$245	\$957	\$925	\$2,127
		DAVID K.	3.8	19	\$325	\$714	\$789	\$1,828
		JOSE K.	3.5	18	\$345	\$755	\$856	\$1,956
		PETER K.	3.6	18	\$352	\$655	\$225	\$1,232
		STEVE L.	3.5	18	\$355	\$788	\$455	\$1,598
		DANIEL M.	2.3	27	\$284	\$766	\$445	\$1,495
		PAUL P.	2.6	19	\$298	\$684	\$898	\$1,880
		NORMAN R.	3.9	19	\$420	\$688	\$898	\$2,006
		STEPHEN S.	3.7	20	\$126	\$544	\$898	\$1,568
		AL X.	2.5	20	\$376	\$855	\$562	\$1,793
MALE	SAM Y.	4.0	24	\$120	\$598	\$256	\$974	
	DORY A.	4.0	20	\$381	\$945	\$1,144	\$2,470	
	REGGIE A.	3.6	20	\$100	\$841	\$654	\$1,595	
	VALERIE A.	3.2	20	\$532	\$756	\$250	\$1,538	
	MARIA E.	3.4	20	\$566	\$485	\$656	\$1,707	
	ERMA F.	1.9	20	\$541	\$755	\$562	\$1,858	
	SARA F.	3.3	20	\$644	\$590	\$224	\$1,458	
	SALLY I.	3.5	20	\$245	\$766	\$544	\$1,555	
	TERRY K.	4.0	24	\$267	\$520	\$454	\$1,241	
	HARRIET P.	3.0	20	\$501	\$688	\$452	\$1,641	
FEMALE	CYNTHIA R.	3.0	19	\$566	\$988	\$620	\$2,174	
	KATE R.	3.5	22	\$226	\$866	\$1,242	\$2,334	
	DONNA S.	3.5	20	\$245	\$700	\$220	\$1,165	
	LORRAINE S	3.2	20	\$572	\$954	\$622	\$2,148	
	ANNA V.	2.3	20	\$599	\$966	\$478	\$2,043	
	MALE	GERALD A.	3.5	20	\$265	\$655	\$687	\$1,607
		GEORGE B.	2.5	21	\$522	\$752	\$425	\$1,699
		BILL D.	3.8	18	\$244	\$810	\$620	\$1,674
		ALLAN E.	3.5	21	\$345	\$455	\$852	\$1,652

The SAS System

ACADEMIC YEAR	SEX	GPA	AGE	\$-BOOKS	\$-FOOD	\$-ENTERTAIN	\$-TOTAL
FRESHMAN	FEMALE	30	161	\$4,279	\$6,541	\$5,080	\$15900
	MALE	38	259	\$5,506	\$9,189	\$9,081	\$23776
SOPHOMORE	FEMALE	43	264	\$4,125	\$9,477	\$8,294	\$21896
	MALE	57	335	\$5,115	\$12034	\$10979	\$28128
JUNIOR	FEMALE	45	285	\$5,985	\$10820	\$8,122	\$24927
	MALE	61	393	\$7,185	\$15963	\$9,860	\$33008
SENIOR	FEMALE	21	143	\$3,031	\$6,281	\$4,696	\$14008
	MALE	26	207	\$5,164	\$6,801	\$5,007	\$16972

**OUTPUT - 4**

PROC REPORT - SAMPLE REPORTS  
----- ACADEMIC SURVEY -----

ACADEMIC YEAR	SEX	MEAN GPA	MEAN AGE	-----ANNUAL \$-BOOKS	MEAN \$-FOOD	AMOUNT \$-ENTERTAIN	SPENT ON \$-TOTAL
FRESHMAN	FEMALE	3.3	18	\$475	\$727	\$564	\$1,767
	MALE	3.2	22	\$459	\$766	\$757	\$1,981
		3.2	20	\$466	\$749	\$674	\$1,889
SOPHOMORE	FEMALE	3.3	20	\$317	\$729	\$638	\$1,684
	MALE	3.3	20	\$301	\$708	\$646	\$1,655
		3.3	20	\$308	\$717	\$642	\$1,667
JUNIOR	FEMALE	3.2	20	\$428	\$773	\$580	\$1,781
	MALE	3.2	21	\$378	\$840	\$519	\$1,737
		3.2	21	\$399	\$812	\$545	\$1,756
SENIOR	FEMALE	3.0	20	\$433	\$897	\$671	\$2,001
	MALE	2.9	23	\$574	\$756	\$556	\$1,886
		2.9	22	\$512	\$818	\$606	\$1,936
		3.2	20	\$404	\$771	\$611	\$1,786

**OUTPUT - 5**

PROC REPORT - SAMPLE REPORTS  
 ----- ACADEMIC SURVEY -----

NAME	ACADEMIC YEAR	GPA	NAME	ACADEMIC YEAR	GPA	NAME	ACADEMIC YEAR	GPA
DORY A.	JUNIOR	4.0	SALLY I.	JUNIOR	3.5	ARLENE T.	FRESHMAN	3.5
GERALD A.	JUNIOR	3.5	WILLIAM I.	SOPHOMORE	3.6	BEN T.	JUNIOR	3.2
HARRY A.	SENIOR	3.6	ALAN J.	JUNIOR	3.4	ELLEN T.	SOPHOMORE	3.0
REGGIE A.	JUNIOR	3.6	MIKE J.	FRESHMAN	3.2	MILLY T.	SOPHOMORE	4.0
VALERIE A.	JUNIOR	3.2	CONNIE K.	SOPHOMORE	2.9	WILL T.	SENIOR	2.9
ANTHONY B.	FRESHMAN	2.0	DAVID K.	SOPHOMORE	3.8	ANNA V.	JUNIOR	2.3
GEORGE B.	JUNIOR	2.5	EILEEN K.	SENIOR	3.0	MANNY V.	FRESHMAN	3.4
TOM B.	SOPHOMORE	3.8	JANE K.	FRESHMAN	2.9	DAVID W.	JUNIOR	2.5
CAROL C.	SENIOR	1.6	JOSE K.	SOPHOMORE	3.5	PAUL W.	SENIOR	3.0
EDWARD C.	FRESHMAN	2.9	MARIE K.	SOPHOMORE	3.2	AL X.	SOPHOMORE	2.5
KATHY C.	SENIOR	2.5	PETER K.	SOPHOMORE	3.6	ROBIN Y.	SOPHOMORE	3.5
BILL D.	JUNIOR	3.8	TERRY K.	JUNIOR	4.0	SAM Y.	SOPHOMORE	4.0
CARMEN D.	SENIOR	3.9	FRANK L.	FRESHMAN	3.5	TONY Y.	SENIOR	2.0
DONNA D.	FRESHMAN	3.2	GEORGE L.	FRESHMAN	2.8	DINA Z.	SOPHOMORE	3.5
FRANK D.	SENIOR	3.6	SHEILA L.	SENIOR	4.0			
JOAN D.	SOPHOMORE	2.0	STEVE L.	SOPHOMORE	3.5			
LEON D.	FRESHMAN	2.9	DAN M.	JUNIOR	1.6			
ROSE D.	SOPHOMORE	3.6	DANIEL M.	SOPHOMORE	2.3			
TOM D.	SOPHOMORE	3.6	JACKIE M.	FRESHMAN	2.1			
ALLAN E.	JUNIOR	3.5	LOU M.	JUNIOR	2.0			
BETTY E.	SOPHOMORE	3.6	FRED O.	SENIOR	2.5			
DENNIS E.	JUNIOR	2.8	JOAN O.	FRESHMAN	2.8			
EVELYN E.	SOPHOMORE	3.2	HARRIET P.	JUNIOR	3.0			
JONAH E.	SENIOR	3.2	PAUL P.	SOPHOMORE	2.6			
LILLY E.	FRESHMAN	3.6	LARRY Q.	JUNIOR	3.7			
MARIA E.	JUNIOR	3.4	CARL R.	JUNIOR	3.6			
SAM E.	FRESHMAN	3.4	CYNTHIA R.	JUNIOR	3.0			
WILLIAM E.	SOPHOMORE	3.7	DEBBIE R.	FRESHMAN	3.8			
CHERYL F.	SOPHOMORE	3.8	HENRY R.	JUNIOR	3.8			
CINDY F.	SOPHOMORE	3.7	JIM R.	FRESHMAN	3.7			
ERMA F.	JUNIOR	1.9	KATE R.	JUNIOR	3.5			
JOHN F.	FRESHMAN	3.8	MICHAEL R.	JUNIOR	3.7			
LAURIE F.	SOPHOMORE	3.2	NORMAN R.	SOPHOMORE	3.9			
OSCAR F.	SOPHOMORE	1.5	SALLY R.	FRESHMAN	3.8			
SARA F.	JUNIOR	3.3	BOB S.	JUNIOR	3.5			
ED G.	JUNIOR	3.5	DONNA S.	JUNIOR	3.5			
MARIE G.	SENIOR	3.2	ERICA S.	FRESHMAN	4.0			
WAYNE G.	SOPHOMORE	3.3	JERRY S.	JUNIOR	3.9			
ED H.	SOPHOMORE	4.0	LORRAINE S	JUNIOR	3.2			
JAMES H.	JUNIOR	3.4	PHILIP S.	SENIOR	3.0			
JOEL H.	FRESHMAN	2.5	ROBERT S.	FRESHMAN	3.8			
RICHARD H.	JUNIOR	3.4	STEPHEN S.	SOPHOMORE	3.7			
TERRY H.	SENIOR	2.0	SUSAN S.	SENIOR	2.8			

PROC REPORT - SAMPLE REPORTS  
 ----- ACADEMIC SURVEY -----

NAME=DORY A.	YEAR=JUNIOR	AGE= 20	SEX=FEMALE	GPA=4.0
\$-BOOKS= \$381	\$-FOOD= \$945	\$-ENTERTAIN=\$1,144	\$-TOTAL= \$2,470	
NAME=GERALD A.	YEAR=JUNIOR	AGE= 20	SEX=MALE	GPA=3.5
\$-BOOKS= \$265	\$-FOOD= \$655	\$-ENTERTAIN= \$687	\$-TOTAL= \$1,607	
NAME=HARRY A.	YEAR=SENIOR	AGE= 19	SEX=MALE	GPA=3.6
\$-BOOKS= \$544	\$-FOOD= \$855	\$-ENTERTAIN= \$478	\$-TOTAL= \$1,877	
NAME=REGGIE A.	YEAR=JUNIOR	AGE= 20	SEX=FEMALE	GPA=3.6
\$-BOOKS= \$100	\$-FOOD= \$841	\$-ENTERTAIN= \$654	\$-TOTAL= \$1,595	
NAME=VALERIE A.	YEAR=JUNIOR	AGE= 20	SEX=FEMALE	GPA=3.2
\$-BOOKS= \$532	\$-FOOD= \$756	\$-ENTERTAIN= \$250	\$-TOTAL= \$1,538	
NAME=ANTHONY B.	YEAR=FRESHMAN	AGE= 16	SEX=MALE	GPA=2.0
\$-BOOKS= \$425	\$-FOOD= \$855	\$-ENTERTAIN= \$965	\$-TOTAL= \$2,245	
NAME=GEORGE B.	YEAR=JUNIOR	AGE= 21	SEX=MALE	GPA=2.5
\$-BOOKS= \$522	\$-FOOD= \$752	\$-ENTERTAIN= \$425	\$-TOTAL= \$1,699	
NAME=TOM B.	YEAR=SOPHOMORE	AGE= 19	SEX=MALE	GPA=3.8
\$-BOOKS= \$215	\$-FOOD= \$625	\$-ENTERTAIN= \$622	\$-TOTAL= \$1,462	
NAME=CAROL C.	YEAR=SENIOR	AGE= 21	SEX=FEMALE	GPA=1.6
\$-BOOKS= \$344	\$-FOOD=\$1,200	\$-ENTERTAIN= \$602	\$-TOTAL= \$2,146	
NAME=EDWARD C.	YEAR=FRESHMAN	AGE= 18	SEX=MALE	GPA=2.9
\$-BOOKS= \$451	\$-FOOD= \$466	\$-ENTERTAIN=\$2,101	\$-TOTAL= \$3,018	
NAME=KATHY C.	YEAR=SENIOR	AGE= 21	SEX=FEMALE	GPA=2.5
\$-BOOKS= \$544	\$-FOOD= \$842	\$-ENTERTAIN= \$874	\$-TOTAL= \$2,260	
NAME=BILL D.	YEAR=JUNIOR	AGE= 18	SEX=MALE	GPA=3.8
\$-BOOKS= \$244	\$-FOOD= \$810	\$-ENTERTAIN= \$620	\$-TOTAL= \$1,674	
NAME=CARMEN D.	YEAR=SENIOR	AGE= 18	SEX=FEMALE	GPA=3.9
\$-BOOKS= \$522	\$-FOOD=\$1,022	\$-ENTERTAIN= \$587	\$-TOTAL= \$2,131	
NAME=DONNA D.	YEAR=FRESHMAN	AGE= 19	SEX=FEMALE	GPA=3.2
\$-BOOKS= \$652	\$-FOOD= \$802	\$-ENTERTAIN= \$785	\$-TOTAL= \$2,239	
NAME=FRANK D.	YEAR=SENIOR	AGE= 21	SEX=MALE	GPA=3.6
\$-BOOKS= \$675	\$-FOOD= \$841	\$-ENTERTAIN= \$254	\$-TOTAL= \$1,770	
NAME=JOAN D.	YEAR=SOPHOMORE	AGE= 27	SEX=FEMALE	GPA=2.0
\$-BOOKS= \$680	\$-FOOD= \$789	\$-ENTERTAIN= \$855	\$-TOTAL= \$2,324	
NAME=LEON D.	YEAR=FRESHMAN	AGE= 18	SEX=MALE	GPA=2.9
\$-BOOKS= \$201	\$-FOOD= \$755	\$-ENTERTAIN= \$400	\$-TOTAL= \$1,356	
NAME=ROSE D.	YEAR=SOPHOMORE	AGE= 18	SEX=FEMALE	GPA=3.6
\$-BOOKS= \$205	\$-FOOD= \$866	\$-ENTERTAIN= \$100	\$-TOTAL= \$1,171	

OUTPUT - 7

PROC REPORT - SAMPLE REPORTS  
 ----- ACADEMIC SURVEY -----

ACADEMIC YEAR	SEX	
	FEMALE GPA	MALE GPA
FRESHMAN	3.3	3.2
JUNIOR	3.2	3.2
SENIOR	3.0	2.9
SOPHOMORE	3.3	3.3

OUTPUT - 8